#301 (Track 25)

This run plays this motive with a sound reminding of a piano.



For this run, 4 kinds of notes are distinguished and treated differently:

1) brief and low notes played on instrument #1 (duration ≤ .2s, frequency ≤ 250Hz).
The amplitude is controlled by F3:





For a duration smaller than about .2s, this function will give a sharp attack; the decay consists of 3 linear portions: the 2 first approximate an exponential shape, the 3rd tries to imitate the effect of a damper. The wave form is given by F1, which consists of 10 harmonics.



With 10Kc sampling rate, all the harmonics will be heard without foldover up to a fundamental frequency of about 400Hz.

2) brief and high notes played on instrument #2 (duration ≤.2s, frequency ≥250Hz).

This instrument is similar 1;o instrument #1 except that the waveshape is given by F2, which consists of only 7 harmonics.



3) long and low notes played on instrument #3. (duration ≥ 2s, frequency ≥ 250Hz).

This instrument is similar to instrument #1, but here the amplitude is controlled by F4.

F4 decays exponentially from 1 to 2^{-6}=1/64. Thus the duration of the note corresponds to 6/10 of "reverberation time" (time for the level to drop 60 dB). In this example, "long" notes last between about .4 and .8s, and this would correspond to a "reverberation time" of the order of 1s, which is shorter than that of a real piano (around 1s at 2000Hz, around 10s at 200Hz). (However, in real pianos the initial decay rate is higher, thus the discrepancy is not as large as it would seem from these data).

4) long and high notes played on instrument #4 (duration ≥ .2s, frequency ≥250Hz).

This instrument is similar to instrument #3, but the waveshape is given by F2, as in instrument #3.

```
COMMENT ------------------------JCR301--------------------------------;
COMMENT MANH. BLUES ON TAPE M1485;
COMMENT:BRIEF NOTES;
COMMENT:LOW NOTES;
INS 0 1;OSC P5 P7 B3 F3 P30;OSC B3 P6 B3 F1 P29;OUT B3
B1;END;
COMMENT:HIGH NOTES;
INS 0 2;OSC P5 P7 B3 F3 P30;OSC B3 P6 B3 F2 P29;OUT B3
B1;END;
COMMENT:LONG NOTES;
COMMENT:LOW NOTES;
INS 0 3;OSC P5 P7 B3 F4 P30;OSC B3 P6 B3 F1 P29;OUT B3
B1;END;
COMMENT:HIGH NOTES;
INS 0 4;OSC P5 P7 B3 F4 P30;OSC B3 P6 B3 F2 P29;OUT B3
B1;END;
SIA 0 4 10000;
COMMENT METRONOME MARKING 150;SV2 0 2 30;SV2 0 30 0 150
15 150;
COMMENT:LO NOTE WAVE;
GEN 0 2 1 .158 .316 1 1 .282 .112 .063 .079 .126 .071 10;
COMMENT:HI NOTE WAVE;
GEN 0 2 2 512 1 .282 .089 .1 .071 .089 .05 7;
COMMENT:SHORT NOTE ENVELOPE;
GEN 0 1 3 512 0 0 1 10 .4 200 .2 400 0 512;
COMMENT:LONG NOTE ENVELOPE;
GEN 0 5 4 -6;
NOT 1 3 1.66 300 104 1;NOT 1 3 1.66 300 175 1;
NOT 1 3 1.66 300 233 1;NOT 1 4 1.66 300 277 1;
```

```
NOT 1 4 1.66 300 330 1;
NOT 1.5 3 1.16 250 207 1;NOT 1.5 4 1.16 250 349 1;
NOT 1.5 4 1.16 250 440 1;NOT 1.5 4 1.16 250 554 1;
NOT 2.66 1 .34 300 104 1;NOT 2.66 2 .34 300 175 1;
NOT 2.66 1 .34 300 233 1;NOT 2.66 2 .34 300 277 1;
NOT 2.66 2 .34 300 330 1;
NOT 3 3 1 400 207 1;NOT 3 4 1 400 349 1;
NOT 3 4 1 400 440 1;NOT 3 4 1 400 554 1;
NOT 4 3 2 250 104 1;NOT 4 3 2 250 147 1;
NOT 4 3 2 250 165 1;NOT 4 3 2 250 196 1;
NOT 4 3 2 250 233 1;
NOT 5 3 1 300 207 1;NOT 5 4 1 300 294 1;
NOT 5 4 1 300 330 1;NOT 5 4 1 300 392 1;
NOT 5 4 1 300 494 1;
NOT 6 3 1.66 300 104 1;NOT 6 3 1.66 300 175 1;
NOT 6 3 1.66 300 233 1;NOT 6 3 1.66 300 277 1;
NOT 6 4 1.66 300 330 1;
NOT 6.5 3 1.16 250 207 1;NOT 6.5 4 1.16 250 349 1;
NOT 6.5 4 1.16 250 440 1;NOT 6.5 4 1.16 250 554 1;
NOT 7.66 1 .34 300 104 1;NOT 7.66 1 .34 300 175 1;
NOT 7.66 1 .34 300 233 1;NOT 7.66 2 .34 300 277 1;
NOT 7.66 2 .34 300 330 1;
NOT 8 3 1 400 207 1;NOT 8 4 1 400 349 1;
NOT 8 4 1 400 440 1;
NOT 8 4 1 400 554 1;
NOT 9 3 2 250 104 1;NOT 9 3 2 250 147 1;
NOT 9 3 2 250 165 1;NOT 9 3 2 250 196 1;
NOT 9 3 2 250 233 1;
NOT 10 3 1 300 207 1;NOT 10 4 1 300 294 1;
NOT 10 4 1 300 330 1;NOT 10 4 1 300 392 1;
NOT 10 4 1 300 494 1;          TER 15;
```



```
      SUBROUTINE CONVT
      COMMON IP(10),P(100)G(1000)
      IF(P(1).NE.1.)GOTO100
      F=511./G(4)
      P(6)=F*P(6)
      P(7)=F/P(4)
  100 RETURN
      END
```

#400 (Track 26)

This run gives a few percussive sounds reminding of a drum--with and without snares. The 1st and the 3rd sections are played back at a sampling rate of 20,000 Hz, and the 2nd section is played back at a sampling rate of 5000Hz, as specified in the score.

Instrument #1 is used to generate these percussive sounds (it is also used in example #410). It is diagrammed on the next page:



This instrument gives a sound which is the sum of a frequency band, of a sine wave and of an inharmonic spectrum.

The frequency band is generated by random amplitude modulation of a sine wave F1. The center frequency is given by V1*, the half bandwidth by V2*. The envelope is given by function F2, which decays exponentially from 1 to 2^{-12}.



F4 is a sine wave--it is the 10th harmonic of the fundamental frequency specified in P6. Thus if P6 = 20, the actual frequency of this sine wave is 200. The envelope is given by function F8, which decays exponentially from 1 to 2^{-8}.

The "inharmonic" spectrum is, in fact, an approximation to an inharmonic spectrum, obtained by playing a wave containing only high order harmonics at a very low frequency. F3 comprises harmonics #10, 16, 22, 23: thus with a fundamental frequency (specified in P6) of 20, this will give component frequencies 200, 320, 440, 460. The envelope is controlled by F2.

The amplitudes for the noise band, the sine wave and the inharmonic spectrum are given respectively by P5, V3* and V4*.

* third-pass variables

Section 1 gives the following pattern played with a snare-like effect given by a noise band centered at 4000Hz and of 3000Hz bandwidth. The sine wave component has frequency 200Hz.



Section 2 gives the following pattern played "without snares": there is no noise band (P5=0). The four pitches correspond to fundamental frequencies of 120, 140, 150, and 160 Hz.

Section 3 gives the following pattern played again with snares.




```
COMMENT ---------------------JCR400-------------------------.
COMMENT:DRUM AND SNARE DRUM ON TAPE M3586 FILES 2 3 4;
COMMENT:TO SKIP FILES;GEN 0 5 1;
COMMENT:SNARE;
COMMENT:4 KC CENTER 3 KC BAND NOISE  200 HZ SINE AND
MEMBRANE SPECTRUM;
COMMENT:20 KC SAMPLING;SIA 0 4 20000;
COMMENT:FOR DRUM;
INS 0 1,OSC P5 P7 B3 F2 P30;RAN B3 V2 B3 P29 P28 P27;
OSC B3 V1 B3 F1 P26;OUT B3 B1;
OSC V3 P7 B4 F2 P25;SET P8;OSC B4 P6 B4 F3 P24;OUT B4 B1;
OSC V4 P7 B5 F8 P23;SET P9;OSC B5 P6 B5 F4 P22;OUT B5
B1;END;
GEN 0 2 1 1 1;
GEN 0 7 2 -12;
GEN 0 4 3 1 10 0 1 512 1.5 16 0 1 512 2 22 0 1 512 1.5 23 0 1 512;
GEN 0 4 4 1 10 0 1 512;
GEN 0 1 7 0 1 .99 5 0 100 0 512;
GEN 0 7 8 -8;
SV3 0 1 100 65 300 800;
NOT .4 1 .2 1000 20 0;
NOT .8 1 .2 1000 20 0;
NOT 1.1 1 .15 1000 20 0;
NOT 1.2 1 .2 1000 20 0;
NOT 1.6 1 .2 1000 20 0;
NOT 1.9 1 .15 1000 20 0;
NOT 2 1 .2 1000 20 0;
NOT 2.4 1 .2 1000 20 0;
NOT 2.8 1 .2 1000 20 0;
NOT 3.1 1 .15 1000 20 0;
```

```
NOT 3.2 1 .2 1000 20 0;
NOT 3.6 1 .2 1000 20 0;
NOT 3.9 1 .15 1000 20 0;
NOT 4 1 .2 1000 20 0;
NOT 4.4 1 .2 1000 20 0;
NOT 4.8 1 .2 1000 20 0;
NOT 5.2 1 .2 1000 20 0;
NOT 5.6 1 .2 1000 20 0;
NOT 6 1 .25 1300 20 0;
SEC 8;
COMMENT:TO WRITE END OF FILE MARK;GEN 0 5 0;
COMMENT:DRUM;COMMENT: 5 KC SAMPLING;SIA 0 4 5000;
COMMENT:MEMBRANE SPECTRUM,SINE WAVE,NO NOISE BAND;
SV3 0 31 7.5 2.5 500 1500;
NOT .4 1 .3 0 12 0;NOT .6 1 .2 0 16 0;
NOT 1.07 1 .2 0 12 0;NOT 1.2 1 .2 0 16 0;
NOT 1.6 1 .3 0 12 0;NOT 2 1 .25 0 14 0;
NOT 2.4 1 .23 0 15 0;
NOT 2.6 1 .27 0 15 0;
NOT 3.07 1 .23 0 15 0;NOT 3.2 1 .23 0 15 0;
NOT 3.6 1 .23 0 15 0;
NOT 4 1 .23 0 15 0;
SEC 6;
COMMENT:TO WRITE END OF FILE MARK;GEN 0 5 0;
COMMENT: SNARE DRUM;COMMENT: 20 KC SAMPLING;SIA 0 4
20000;
GEN 0 7 2 -12;
GEN 0 1 7 0 1 .99 5 0 100 0 512;
GEN 0 7 8 -8;
SV3 0 1 100 65 300 800;
NOT .4 1 .15 1000 20 0;NOT .6 1 .2 1000 20 0;
```



```
NOT 1.07 1 .2 1000 20 0;
NOT 1.2 1 .2 1000 20 0;
NOT 1.6 1 .2 1000 20 0;
NOT 2.0 1 .2 1000 20 0;
NOT 2.4 1 .25 1200 20 0;NOT 2.9 1 .15 1000 20 0;
NOT 3 1 .15 1000 20 0;
NOT 3.1 1 .15 1000 20 0;
NOT 3.2 1 .20 1000 20 0;
NOT 3.55 1 .15 700 20 0;
NOT 3.6 1 .2 700 20 0;
NOT 4 1 .15 800 20 0;
NOT 4.06 1 .15 800 20 0;
NOT 4.13 1 .15 800 20 0;
NOT 4.20 1 .15 800 20 0;
NOT 4.27 1 .15 800 20 0;
NOT 4.33 1 .15 800 20 0;
NOT 4.4 1 .22 1200 20 0;
TER 6;

      SUBROUTINE CONVT
      COMMON IP(10),P(100),G(1000)
      IF(P(1).NE.1.)GOTO100
      F=511./G(4)
      P(6)=F*P(6)
      P(7)=F/P(4)
      IF(P(3).EQ.1.)GOTO100
      P(6)=P(6)*G(11)
100   RETURN
      END
```

#410 (Track 27)

This run gives a few percussive sounds. The two first sections are played back at a sampling rate of 5000Hz, while they have been synthesized with a specified rate of 10,000Hz: hence, for these two sections, the durations are the double and the frequencies are the half of those specified in the score. The two last sections, giving two bell-like sounds, are played back at sampling rate 10,000Hz and the frequencies and durations are as specified in the score.

Instrument #1 is used to generate the percussive sounds of the two first sections. It is diagrammed below:



This instrument gives a sound which is the sum of a frequency band and of an inharmonic spectrum. This sound decays exponentially.

The frequency band is generated by random amplitude modulation of a sine wave F1. The center frequency is given by V1, the half-bandwidth by V2.* The envelope is controlled by function F2, which decays exponentially from 1 to 2^{-7}.

* The values of 3rd pass variables V1 & V2 correspond to a center frequency and a bandwidth of 1000 and 800 Hz at 10Kc sampling rate, hence of 500 & 400 Hz in the sound example, recorded at 5Kc.



The "inharmonic" spectrum is actually harmonic: an approximation to an inharmonic spectrum is obtained by playing at a very low frequency a wave containing only high order harmonics. For instance, spectrum 1 is generated by periodic waves comprising harmonics #10, #16, #22, #23, #25, #29, #32: at frequency 10 (specified by P6), this will give component frequencies 100, 160, 220, 230, 250, 290, 320. Spectrum 1 is here obtained by the sum of functions F3, comprising harmonics 10, 16, 22, 23, and F4, comprising harmonics 25, 29, 32: the envelope of F3 is controlled by F2, whereas the envelope of F4 is controlled by F8, which insures a faster decay for the higher components.



By means of P7 and P9, functions F3 and F4 can be changed to other functions. Here, in addition to spectrum 1, which is an approximation to a membrane spectrum, an approximation to the spectrum of a struck metallic object has been tried: it is called spectrum 2. It is obtained by the sum of functions F5, which comprises harmonics 16, 20, 22, 34, 38, 47, and F6, which comprises harmonics 50, 53, 65, 70, 75, 77, 100.

Instrument #2, diagrammed below, is used to generate inharmonically related frequency components of bell-like sounds. The waveshape F1 is a sine wave. Function F7, controlling the envelope, decays exponentially from 1 to 2^{-7}. The lowest component frequency is specified in the score by V11 (Pass II variable); in a component note card, P6 specifies the ratio of the frequency of a component to the

frequency of the lowest component: CONVT multiplies P6 by V11. (E.g., if the components were harmonically related, the P6 would be 1, 2, 3 . . .).



Section 1 includes 3 sounds of actual duration .8, 2, and 4s played on instrument #1, first with spectrum 1, then with spectrum 2, with a fundamental frequency of 50Hz.

Section 2 includes 3 sounds of duration 18, 2, and 4s played on instrument #1 with spectrum #2, with a fundamental frequency of 150Hz.

Section 3 gives a bell-like sound played with instrument #2. It consists of 7 components of frequencies proportional to 1, 2, 2.4, 3, 4.5, 5.33, 6 having different decay times. The lowest component is at frequency 329.

Section 4 gives a bell-like sound played with instrument #2, consisting of only 4 components.

```
COMMENT-----------------------JCR410---------------------;
COMMENT:PERCUSSION;
COMMENT:TAPE M1172;
COMMENT:FOR DRUM;
INS 0 1;OSC P5 P7 B3 F2 P30;RAN B3 V2 B3 P29 P28 P27;
OSC B3 V1 B3 F1 P26;OUT B3 B1;
OSC V3 P7 B4 F2 P25;SET P8;OSC B4 P6 B4 F3 P24;OUT B4 B1;
OSC V4 P7 B5 F8 P23;SET P9;OSC B5 P6 B5 F4 P22;OUT B5
B1;END;
SV3 0 1 50 40 500 500;
COMMENT:FOR BELLS;
INS 0 2;OSC P5 P7 B3 F7 P30;OSC B3 P6 B3 F1 P29;OUT B3
B1;END;
GEN 0 2 1 1 1;GEN 0 7 2 -7;
COMMENT: FOR SPECTRUM 1;
GEN 0 4 3 1 10 0 1 512 1.5 16 0 1 512 2 22 0 1 512 1.5 23 0 1 512;
GEN 0 4 4 1 25 0 1 512 .5 29 0 1 512 .2 32 0 1 512;
COMMENT: FOR SPECTRUM 2;
GEN 0 4 5 1 15 0 1 512 1 20 0 1 512 1 22 0 1 512 2 34 0 1 512
 1 38 0 1 512 1 1 47 0 1 512;
GEN 0 4 6 512 2 50 0 1 512 1 53 0 1 512 1 65 0 1 512 1 70 0 1 512
 75 0 1 512 1 77 0 1 512 1 100 0 1 512;
GEN 0 7 7 -8;GEN 0 7 8 -12;
COMMENT:DRUM;
COMMENT:FREQUENCY 100 HZ AT 10 KC SAMPLING RATE;
COMMENT:SPECTRUM 1;
NOT 1 1 .4 500 10 0;NOT 2 1 1 500 10 0;
NOT 3.5 1 2 500 10 0;
COMMENT:SPECTRUM 2;
NOT 6 1 .4 500 10 0 5 6;NOT 7 1 1 500 10 0 5 6;
NOT 8.5 1 2 500 10 0 5 6;
```



```
SEC 12;
COMMENT: FREQUENCY 300 HZ SPECTRUM 2;
NOT 1 1 .4 500 30 0;NOT 2 1 1 500 30 0;
NOT 3.5 1 2 500 30 0;
SEC 6;
COMMENT:BELL LIKE SOUNDS;
COMMENT:LOWEST FREQUENCY 329;        SV2 0 11 329;
NOT 1 2 3 200 1 0;NOT 1 2 2.8 200 2 0;NOT 1 2 2.7 200 2.4  0;
NOT 1 2 2.4 200 3 0;NOT 1 2 2.2 200 4.5 0;NOT 1 2 3.2 400 2.5 0;
NOT 1 2 1.5 300 6 0;
SEC 5;
NOT 1 2 4 400 1 0;NOT 1 2 3.5 400 2 0;NOT 1 2 3.2 400 2.5 0;
NOT 1 2 2.9 400 3.36 0;
TER 7;
```

```
      SUBROUTINE CONVT
      COMMON IP(10),P(100),G(1000)
      IF(P(1).NE.1.)GOTO100
      F=511./G(4)
      P(6)=F*P(6)
      P(7)=F/P(4)
      IF(P(3).EQ.1.)GOTO100
      P(6)=P(6)*G(11)
100   RETURN
      END
```

#411 (Track 28)

This run gives some more percussive sounds, with the instruments and functions described for #410. Here the sampling rate for playback is 10,000Hz, as specified in the score.

1st section gives 6 sounds of increasing pitches played on instrument #1 with spectrum 1.

2nd section is similar to 1st section, but with spectrum 2.

3rd section gives 3 sounds of increasing durations and decreasing pitches with spectrum 1.

4th section is similar to 3rd section, but with spectrum 2.

5th, 6th, and 7th sections give 4 bell-like sounds played on instrument #2.

```
COMMENT --------------------------JCR411-------------------------------;
COMMENT:PERCUSSION
COMMENT:FOR DRUM;
INS 0 1;OSC P5 P7 B3 F2 P30;RAN B3 V2 B3 P29 P28 P27;
OSC B3 V1 B3 F1 P26;OUT B3 B1;
OSC V3 P7 B4 F2 P25;SET P8;OSC B4 P6 B4 F3 P24;OUT B4 B1;
OSC V4 P7 B5 F8 P23;SET P9;OSC B5 P6 B5 F4 P22;OUT B5
B1;END;
SV3 0 1 50 40 500 500;
COMMENT:FOR BELLS;
INS 0 2;OSC P5 P7 B3 F7 P30;OSC B3 P6 B3 F1 P29;OUT B3
B1;END;
GEN 0 2 1 1 1;GEN 0 7 2 -7;
COMMENT: FOR SPECTRUM 1;
GEN 0 4 3 1 10 0 1 512 1.5 16 0 1 512 2 22 0 1 512 1.5 23 0 1 512;
GEN 0 4 4 1 25 0 1 512 .5 29 0 1 512 .2 32 0 1 512;
COMMENT: FOR SPECTRUM 2;
```



```
GEN 0 4 5 1 15 0 1 512 1 20 0 1 512 1 22 0 1 512 2 34 0 1 512
 1 38 0 1 512 1 1 47 0 1 512;
GEN 0 4 6 512 2 50 0 1 512 1 53 0 1 512 1 65 0 1 512 1 70 0 1 512
 75 0 1 512 1 77 0 1 512 1 100 0 1 512;
GEN 0 7 7 -8;GEN 0 7 8 -12;
COMMENT:DRUM;
COMMENT:BRIEF SOUNDS SPECTRUM 1;
NOT 1.2 1 .5 150 5.8 3 3 4;
NOT 1.8 1 .2 150 9.8 0;
NOT 2 1 .4 150 13.9 0;
NOT 2.9 1 .3 150 22 0;
NOT 3.5 1 .3 150 31.1 0;
NOT 3.8 1 .2 150 49.3 0;
SEC 5;
COMMENT:BRIEF SOUNDS SPECTRUM 2;
NOT 1.2 1 .5 150 5.8 3 5 6;
NOT 1.8 1 .2 150 9.8 0;
NOT 2 1 .4 150 13.9 0;
NOT 2.9 1 .3 150 22 0;
NOT 3.5 1 .3 150 31.1 0;
NOT 3.8 1 .2 150 49.3 0;
SEC 5;
COMMENT:LONGER SOUNDS SPECTRUM 1;
NOT 0 1 1 150 22 0 3 4;
NOT 1 1 2 500 13.9 0;
NOT 3 1 4 800 5.8 0;
SEC 10;
COMMENT:LONGER SOUNDS SPECTRUM 2;
NOT 0 1 1 150 22 0 5 6;
NOT 1 1 2 500 13.9 0;
NOT 3 1 4 800 5.8 0;
```

```
SEC 10;
COMMENT:BELLS;
COMMENT:FREQUENCE FALE 104;SV2 0 11 104;
NOT 1 2 2 400 1 0;NOT 1 2 1.5 400 1.5 0;NOT 1 2 1.5 400 2 0;
NOT 1 2 1.3 400 2.7 0;NOT 1 2 1.1 400 3.3 0;
NOT 5 2 3 200 1 0;
NOT 5 2 2.8 200 1.65 0;
NOT 5 2 2.7 200 2.10 0;
NOT 5 2 2.4 200 3 0;
NOT 5 2 2.1 200 3.54 0;NOT 5 2 2 200 4.97 0;NOT 5 2 1.5 200 5.33 0;
SEC 10;
NOT 1 2 3 200 1 0;NOT 1 2 2.8 200 2 0;NOT 1 2 2.7 200 2.4 0;
NOT 1 2 2.4 200 3 0;NOT 1 2 2.2 200 4.5 0;NOT 1 2 2 300 5.33 0;
SEC 5;
NOT 1 2 4 400 1 0;NOT 1 2 3.5 400 2 0;NOT 1 2 3.2 400 2.5 0;
NOT 1 2 2.9 400 3.36 0;
TER 7;

      SUBROUTINE CONVT
      COMMON IP(10),P(100),G(1000)
      IF(P(1).NE.1.)GOTO100
      F=511./G(4)
      P(6)=F*P(6)
      P(7)=F/P(4)
      IF(P(3).EQ.1.)GOTO100
      P(6)=P(6)*G(11)
  100 RETURN
      END
```


#420 (Track 29)

This run gives percussive sounds reminding of gong sounds. There is a separate note card for each frequency component of the sound; all components are generated by instrument #1, diagrammed below



The waveshape F1 is a sine wave. Function F2 controls the envelope; F2 is decaying exponentially from 1 to 2^{-7}. The component frequency is given by P6*, its initial amplitude by P5 and the duration of the decay by P4*, duration of the note.

The frequencies of the components are not harmonically related.

In the first sound, all frequency components decay synchronously. The spectrum is thus invariant; the effect reminds of an element of an electronic chime.

In the second sound, the same frequency components have a decay time approximately inversely proportional to their frequencies (although this principle is not followed inflexibly, to give a more intricate decay pattern). The sound has more life and naturalness than the first one.

The following sound consists of different frequency components with non-synchronous decay.

Then follow four partly overlapping sounds of the same type; the beating of close components gives some warmth to the sound.

* This run has been computed with a sampling rate of 20,000Hz, but the sound example presents it played back with a sampling rate of 5000Hz. Hence actual durations correspond to 4 times the values indicated in P4; actual frequencies correspond to .25 times the values indicated in P5.

```
COMMENT ------------------JCR420--------------------;
COMMENT:GONG LIKE SOUNDS;
COMMENT:RUN 2 ON TAPE M3282 FILE 2;GEN 0 5 1;
INS 0 1;OSC P5 P7 B3 F2 P30;OSC B3 P6 B3 F1 P29;OUT B3
B1;END;
COMMENT:ORIGINAL SAMPLING RATE 20000 HZ;    SIA 0 4 20000;
COMMENT: PLAY AT A SAMPLING RATE OF 5000 HZ;
```



```
COMMENT:HENCE DURATIONS MULTIPLIED BY 4, FREQUENCIES
DIVIDED BY 4;
GEN 0 2 1  1 1;GEN 0 7 2 512 -7;
COMMENT:FOR DEMONSTRATION FIRST NOTE WITH
SYNCHRONOUS DECAY;
 NOT 1 1 2.5 300 960 0; NOT 1 1 2.5 250 1110 0; NOT 1 1 2.5 200 1540
0;
 NOT 1 1 2.5 300 2420 0; NOT 1 1 2.5 100 1360 0; NOT 1 1 2.5 100
2680 0;
 NOT 1 1 2.5 100 3250 0;
SEC 5;
 NOT 1 1 2.5 300 960 0; NOT 1 1 2.4 250 1110 0; NOT 1 1 2.2 200 1540
0;
 NOT 1 1 0.4 300 2420 0; NOT 1 1 2 100 1360 0; NOT 1 1 1.3 100 2680
0;
 NOT 1 1 1 100 3250 0;
 NOT 5 1 2 300 970 0;NOT 5 1 1.9 250 1230 0;NOT 5 1 1.7 100 1360 0;
 NOT 5 1 1.2 200 1536 0;NOT 5 1 .9 100 2048 0;NOT 5 1 .7 150 3280 0;
SEC 8;
NOT 1 1 2.5 150 960 0;NOT 1 1 2.4 125 1110 0;NOT 1 1 2.2 150 1540
0;
NOT 1 1 0.8 100 2420 0;NOT 1 1 2 50 1360 0;NOT 1 1 1.3 50 2680 0;
NOT 1 1 1 50 3250 0;
NOT 1.7 1 2.2 200 965 0;NOT 1.7 1 2.1 150 1050 0;NOT 1.7 1 1 250
1430 0;
NOT 1.7 1 1.3 100 1210 0;NOT 1.7 1 1.1 100 1260 0;
NOT 1.7 1 1.9 100 1540 0;NOT 1.7 1 1.6 100 1930 0;
NOT 1.8 1 2.9 300 970 0;NOT 1.8 1 2.7 250 1230 0;
NOT 1.8 1 2.6 100 1360 0;NOT 1.8 1 1.6 200 1536 0;
NOT 1.8 1 1.2 100 2048 0;NOT 1.8 1 1.1 150 3280 0;
NOT 3.2 1 3.4 150 960 0;NOT 3.2 1 3.2 125 1110 0;
```

```
NOT 3.2 1 3 150 1540 0;NOT 3.2 1 2.1 50 2420 0;
NOT 3.2 1 .8 100 1360 0;NOT 3.2 1 1.6 50 2680 0;
NOT 3.2 1 1.1 50 3250 0;
TER 8;
```

```
      SUBROUTINE CONVT
      COMMON IP(10),P(100),G(1000)
      IF(P(1).NE.1.)GOTO100
      F=511./G(4)
      P(6)=F*P(6)
      P(7)=F/P(4)
 100  RETURN
      END
```

#430 (Track 30)

This run gives three successive approximations of a bell sound.

There is a separate note card for each frequency component of the sound; all components are generated by instrument #1, diagrammed below.



The waveshape F1 is a sine wave. Function F2 controls the envelope; F2 is decaying exponentially from 1 to 2^{-10}. The component frequency is given by P6, its amplitude by P5, and the duration of the decay by P7 (P(7) = P(4)).

The frequencies of the components do not form a harmonic series; however, they are not arbitrarily inharmonic. In most actual bells it is attempted to approximate the following ratios for the 1st 5 components: .5, 1, 1.2, 1.5, 2 (corresponding for example to the following succession of notes: G, G, B flat, D, G, called respectively hum notes, fundamental, minor third, fifth, nominal). Here the frequency ratios of the components are as follows: .56, .92, 1.19, 1.71, 2, 2.74, 3, 3.76, 4.07.

In the first sound, all these frequency components decay synchronously. This gives an unnatural sound.

In the second sound, the components have a decay time approximately inversely proportional to their frequencies (although this principle is violated in one instance where a lower component decays faster: this gives a slight bounce a little after the beginning of the sound). The sound is much more natural, yet still a little dull.

In the third sound, each of the two lowest partials is split into two components of slightly different frequencies (224 and 225, 368 and 369.7). This causes beats which add some life and warmth to the sound. It is likely that in real bells partials are split into two close components, due to departure from rotational symmetry.

```
COMMENT---------------JCR430---------------------;
COMMENT:BELL EXPERIMENTS;
COMMENT:ON TAPE M1485 FILE 4;GEN 0 5 3;
COMMENT:5 KC SAMPLING RATE; SIA 0 4 5000;
INS 0 1; OSC P5 P7 B3 F2 P30;OSC B3 P6 B3 F1 P29;
OUT B3 B1;END;
COMMENT:TO SET GENERAL CONVT;     SV2 0 10 2 6 -7;
COMMENT: SYNCHRONOUS DECAY;
NOT 1 1 20 250 224.5 20;NOT 1 1 20 400 368.5 20;
NOT 1 1 20 400 476 20;NOT 1 1 20 250 684 20;
NOT 1 1 20 220 800 20;NOT 1 1 20 200 1096 20;
NOT 1 1 20 200 1200 20;NOT 1 1 20 150 1504 20;
NOT 1 1 20 200 1628 20;SEC 21;
```



```
COMMENT:  NON SYNCHRONOUS DECAY;
NOT 1 1 20 250 224 20;NOT 1 1 12 400 368.5 12;
NOT 1 1 6.5 400 476 6.5;NOT 1 1 7 250 680 7;
NOT 1 1 5 220 800 5;NOT 1 1 4 200 1096 4;
NOT 1 1 3 200 1200 3;NOT 1 1 20 150 1504 2;
NOT 1 1 1.5 200 1628 1.5;
SEC 21;
COMMENT:  NON SYNCHRONOUS DECAY AND TWO SPLIT
PARTIALS;
NOT 1 1 20 1500 224 20;NOT 1 1 18 1000 225 18;
NOT 1 1 13 1500 368 13;NOT 1 1 11 2700 369.7 11;
NOT 1 1 6.5 4000 476 6.5;NOT 1 1 7 2500 680 7;
NOT 1 1 5 2200 800 5;NOT 1 1 4 200 1096 4;
NOT 1 1 3 200 1200 3;NOT 1 1 2 150 1504 2;
NOT 1 1 1.5 200 1628 1.5;
TER 22;
```

#440 (Track 31)

This run gives some drum-like sounds with variable frequency (plus a non drum-like sound).

The sounds are generated by instrument #3, which uses among its inputs the output of degenerate instrument #2.

Instrument #2, diagrammed below, is used to effect pitch changes. It is a degenerate instrument, the output of which goes into B5.



Function F2 controls pitch evolution. P6 gives the duration of the frequency cycle (which for all examples of this run coincides with the note duration-- in fact it is made slightly longer to be sure to avoid a recycling of the frequency function at the end of the note. This can happen due to round off errors in the increment value, especially with computers of 24 bit word length). P5 = 1.

Instrument #3 comprises 3 parallel oscillators with different envelope controls, as shown by the following diagram.



One of these oscillators generates the fundamental of maximum frequency 160Hz (F3 is a sine wave). The amplitude is controlled by F6, sketched below. In all examples given here (except the last note) the attack time is 10ms or 30ms (note this is not a linear attack--otherwise these times would be smaller), the "steady" state lasts 10ms or 30ms and the decay time is about 1.6s.



The two other oscillators play waveshapes F4 and F5, which comprise high order harmonics of a low fundamental--in order to imitate an inharmonic set of partials. (F4 comprises harmonics 3, 4, 5, 6: with P11 = 75, this oscillator will give frequencies 225, 300, 375, 450, similarly F5 comprises harmonics 8, 9, 10, 11, 12, 15, 17, 18 with P16 = 61, which yields frequencies between 500 and 1100). The amplitudes are controlled respectively by F7 and F8, which insure a fast decay for waves F4 and F5 (in this example, about .6 and .3s to decay to 1/1000 of the initial amplitude for all notes except the last one).





The 1st section plays 2 notes with constant pitch--the 2nd note has a longer attack and a 30ms steady state for the fundamental.

The 2nd section is similar, but the pitch is going up a minor third from the beginning to the end of each note.

The 3rd section is similar, but the pitch is going up then down during each note, since the frequency is controlled by function F2 as drawn below.



The 4th section is similar, but the pitch is going down a minor third from the beginning to the end of each note.

The last section gives a note generated with the same instrument but with parameters differing very much from the previous ones, especially a .9s attack time for wave F5 and an attack time occupying practically all the duration (2s) of the note for wave F4. This is only to show how easily a computer instrument designed for a particular purpose can be used to give different types of sound.

```
COMMENT:----------------JCR440--------------------;
COMMENT:VARIABLE PITCH DRUMS;
SIA 0 4 5000;
COMMENT:FOR PITCH VARIATION;
INS 0 2;OSC P5 P6 B5 F2 P30;END;
COMMENT:FOR 3 COMPONENTS;
INS 0 3;ENV P5 F6 B2 P7 P8 P9 P30;
```

```
MLT P6 B5 B6;OSC B2 B6 B2 F3 P29;OUT B2 B1;
ENV P10 F7 B3 P12 P13 P14 P28;
MLT P11 B5 B5;OSC B3 B7 B3 F4 P27;OUT B3 B1;
ENV P15 F8 B4 P17 P18 P19 P26;
MLT P16 B5 B8;OSC B3 B8 B4 F5 P25;OUT B4 B1;END;
COMMENT:TO SET GENERAL CONVT;
SV2 0 20 1 -6;
SV2 0 30 6 6 107 11 112 16 117;
GEN 0 2 3 1 1;
GEN 0 2 4 0 0 1 1 .3 .2 6;
GEN 0 4 5 10 8 0 1 512 8 9 0 1 512 5 10 0 1 512
2 11 0 1 5112 3 12 0 1 512 3 15 0 1 512 2 17 0 1 512 1 18 0 1 512;
COMMENT:FOR ENVELOPE;
GEN 0 6 6 10 .99 .9 10;
GEN 0 6 7 8 .99 .99 24;
GEN 0 6 8 8 .99 .99 40;
COMMENT:CONSTANT PITCH;  GEN 0 1 2 .99 1 .99 512;
NOT 1 2 1.63 1 1.63;
NOT 1 3 1.62 1000 160 .010 0 1.6 600 75 .010 0 1.61 300 61 .010 0
1.61;
NOT  3 2 1.7 1 1.7;
NOT 3 3 1.66 1000 160 .030 0 1.6 600 75 .010 0 1.65 300 61 .010 0
1.65;
SEC 5;
COMMENT:UP A MINOR 3D .85 1;  GEN 0 1 2 .85 1 .99 512;
NOT 1 2 1.63 1 1.63;
NOT 1 3 1.62 1000 160 .010 0 1.6 600 75 .010 0 1.61 300 61 .010 0
1.61;
NOT  3 2 1.7 1 1.7;
NOT 3 3 1.66 1000 160 .030 0 1.6 600 75 .010 0 1.65 300 61 .010 0
1.65;
```



```
SEC 5;
COMMENT:OSCILL PITCH;  GEN 0 2 2 .1 .9 1;
NOT 1 2 1.63 1 1.63;
NOT 1 3 1.62 1000 160 .010 0 1.6 600 75 .010 0 1.61 300 61 .010 0
1.61;
NOT  3 2 1.7 1 1.7;
NOT 3 3 1.66 1000 160 .030 0 1.6 600 75 .010 0 1.65 300 61 .010 0
1.65;
SEC 5;
COMMENT:DOWN A MINOR 3D .85 1;  GEN 0 1 2 .99 1 .85 512;
NOT 1 2 1.63 1 1.63;
NOT 1 3 1.62 1000 160 .010 0 1.6 600 75 .010 0 1.61 300 61 .010 0
1.61;
NOT  3 2 1.7 1 1.7;
NOT 3 3 1.66 1000 160 .030 0 1.6 600 75 .010 0 1.65 300 61 .010 0
1.65;
SEC 5;
COMMENT:NOTE WITH NON REALISTIC PARAMETERS;
NOT 1 2 2 1 2;
NOT 1 3 2 1000 160 .010 0 1.95 600 75 92 0 1.9 800 61 .9 0 .8;
TER 4;
```

#490 (Track 32)

This example presents a fragment obtained through mixing from runs #200, 301, 400, 410, and three other runs.

Three of the original sounds (excerpted from #200 and #410) underwent transposition by speed changing before mixing, the others did not undergo electroacoustic modification (except of course amplitude control). Some tape splicing was involved to excerpt single sounds from

#200 and #410 and to place each element at the proper time. A chart of the beginning of the mixing is given.

As can be heard, the synchronization is not bad; with good tape recorders, it seems easy most of the time to achieve satisfactory synchronization up to durations of 30s to 1 mn. In connection with this, it should be noted that tape recorder speeds often go down substantially, due to changes in tape tension, when one approaches the end of a reel (this has been studied by F. Harvey and J. McLean).

The runs used in this episode and not presented among the previous examples are briefly described below:

(1) a cluster of sinusoids, forming the following chord:



together with two brief episodes played by a simple instrument with feedback (c.f.,#510), and noted as follows:



(2) a run analogous to #301, but where the spectra are gradually moved from a low region (below around 600Hz) to a higher region (between

about 500 and 2500Hz) by means of redefining the functions giving the waveshape in the course of the run;

(3) a run analogous to the 2nd section #410, but with a lower pitch (frequencies about twice lower) and a regular beat:

The remarks mentioned for #512 apply to this example.





#500 (Track 33)

This run presents what might be called a "spectral analysis of a chord": for each note of the chord, successive harmonics are gradually introduced. This is performed automatically by subroutine PLF3, listed with the score and described below. The example is in stereo, with a sampling rate of 20,000Hz for each channel; it is played backwards, because it was desired to terminate on the fundamental notes of the chord. (This can be done also by using negative values for TS). PLF3 is a first pass subroutine, called by the following data statement:

P(1)	P(2)	P(3)	P(4)	P(5)	P(6)	P(7)	P(8)
PLF	Action Time	3	NC	N	TS	FACT	DD

It operates on a number of subsequent note cards, and this number is specified by NC: e.g., if NC=4, PLF3 will operate on the 4 note cards following the PLF data statement. The instrument number has to be 1 or 2, and these instruments must be such that P(6)* gives the note frequency F. PLF3 will add to each note card it operates on N note cards of frequencies 2F, 3F, ..., (N+1)F, played alternately by instrument number 1 and 2. If the action time of the original note card is AT, the action times of the added note cards will be, respectively, AT + TS, AT + 2TS, ..., AT + NTS. In examples #500 and #501, the instrument 1 and 2 give the same tone quality respectively in the left and the right channel. This alternation between instruments can be used as well, for instance, to get alternate harmonics of different timbres or intensities.

PLF3 provides for a multiplication of P(5) by FACT from one harmonic to the next. (If FACT ≤ 0, P(5) is left the same.) This can be

* From now on, the P fields refer to note cards P fields--the P fields of the PLF3 data statement are referred to as NC, N, TS, FACT, DD.

used for example to increase (or to reduce) the amplitude by a constant factor from one harmonic to the next.

Finally, the successive harmonics note durations are related to the fundamental note duration D by D-DD. If DD=0, they have the same duration as the fundamental, as in the figure:



The total duration of the sound is given in this case by D + N x TS. If DD=TS, the pattern is as on the figure.



In this case the total duration of the note is equal to D. Care must be taken to avoid negative durations if DD > O. (DD can as well be negative, to give harmonics lasting longer than the fundamental).

In example #500, PFL3 is applied to the note of a chord noted:

A different PLF statement is used for each pair of bracketed notes: 4 harmonics of group 1, 8 harmonics of group 2 and 10 harmonics of group 3 are generated--at different rates, such that the overall duration is the same for all groups. (Actually the very end of the sound--which becomes the beginning since the example is played backwards--has been cut out). All notes are played by instrument #1 or #2, which are identical, except that 1 plays into the left channel and 2 into the right channel. Instrument 1 is diagrammed below. F1 is a sine wave. This instrument gives a parabolic attack and decay, since F2 is a linear attack and decay as drawn here, and since the output of the amplitude controlling generator is multiplied by itself (which yields the dotted curve).




```
COMMENT---------------------------JCR500------------------------------;
COMMENT:SPECTRAL ANALYSIS OF A CHORD TO BE PLAYED
BACKWARDS;
COMMENT:TAPE M1084 FILE 2;COMMENT :TO SKIP FIRST
FILE;GEN 0 5 1;
COMMENT:G SHARP D NATURAL E B A SHARP  USE SPECIAL
PLF3;
COMMENT:PARABOLIC ATTACK AND DECAY  SAMPLING RATE
20000;SIA  0 4 20000;
INS 0 1;OSC P5 P7 B3 F2 P35;MLT B3 B3 B4;OSC B4 P6 B4 F1 P34;
STR B4 V1 B1;END;
INS 0 2;OSC P5 P7 B3 F2 P35;MLT B3 B3 B4;OSC B4 P6 B4 F1 P34;
STR V1 B4 B1;END;
GEN 0 2 1 1 1;GEN 0 3 2 0 2 4 6 8 10 0;
COMMENT:G SHARP D NATURAL E B A SHARP;
PLF 1 3 2 10 1 0 0;NOT 1 1 2 500 208 2;NOT 1.01 2 2 500 294 2;
```



```
PLF 1 3 2 6 1.25 0 0;NOT 1 1 3.75 500 392 3.75;NOT 1.01 1 3.75 500
659 3.75;
PLF 1 3 2 4 2.5 1;NOT 1 1 7.5 500 988 7.5;NOT 1 1 7.5 500 1865 7.5;
TER 15;

      SUBROUTINE CONVT
      COMMON IP(10),P(100),G(1000)
      IF(P(1).NE.1.)GOTO100
      F=511./G(4)
      P(5)=SQRT(P(5))
      P(6)=F*P(6)
      P(7)=F/P(7)
 100RETURN
      END
CPLF3LB1                     PLF3 FOR LB1
C     GENERATES HARMONICS WITH ALTERNATING
INSTRUMENTS
C     OPERATES ON NOTE CARDS OF INSTS 1 AND 2
C     P(5) AMPLITUDE,P(6) FREQUENCY ON NOTE CARDS
C     ON PLF CARD, P(4) SPECIFIES HOW MANY FOLLOWING NOTE
CARDS
C     WILL BE OPERATED ON
C     P(5) GIVES THE NUMBER OF HARMONICS GENERATED
C     P(6) SPECIFIES THE TIME SEPARATION BETWEEN
HARMONICS
C     P(7) SPECIFIES THE AMPLITUDE MULTIPLIER FROM ONE
HARMONIC TO THE
C     NEXT
C     P(8) GIVES THEE DURATION DIMINUTION FROM ONE
HARMONIC TO THE NEXT
      SUBROUTINE PLF3
```

```
      COMMON IP(10),P(100),D(2000)
      NC=P(4)
      N=P(5)
      IS=P(6)
      FACT=P(7)
      DD=P(8)
      DO 1 I=1,NC
      CALL READ1
      CALL WRITE1(10)
      F=P(6)
      DO 2 J=1,N
      P(6)=FLOAT(J+1)*F
      P(2)=P(2)+TS
C     TO CHANGE INSTS NUMBER FROM 1 TO 2 AND VICE VERSA
      AINST=P(3)-1.
      IF(AINST)3,3,4
 3    P(3)=2.
      GOTO5
 4      P(3)=1.
 5    CONTINUE
      IF(FACT.GT.0.) P(5)=P(5)*FACT
      P(4)=P(4)-DD
 2    CALL WRITE1(10)
 1    CONTINUE
100RETURN
      END
```

#501 (Track 34)

This run is similar to #500: the same harmonics from notes of the same chord have been generated by PLF3 (c.f., #500) (except that a

longer portion has been removed from the end of the sound--which again becomes the beginning since the example is played backwards). The difference in tone quality is due to the difference in the envelope of each component: instead of a gradual parabolic attack and decay, each harmonic (for the example played backwards) has an instantaneous attack and an exponential decay, controlled by F7.

```
COMMENT---------------JCR501------------------------;
COMMENT:SPECTRAL ANALYSIS OF A CHORD TO BE PLAYED
BACKWARDS;
COMMENT:G SHARP D NATURAL E B A SHARP  USE SPECIAL
PLF3;
COMMENT:INSTANTANEOUS ATTACK EXPONENTIAL DECAY;
INS 0 1;OSC P5 P7 B3 F2 P35;OSC B3 P6 B4 F1 P34;STR B4 V1
B1;END;
INS 0 2;OSC P5 P7 B3 F2 P35;OSC B3 P6 B4 F1 P34;STR V1 B4
B1;END;
GEN 0 2 1 1 1;GEN 0 7 2 6;
PLF 1 3 2 10 1 0 0;NOT 1 1 2 500 208 2;NOT 1.01 2 2 500 294 2;
PLF 1 3 2 6 1.25 0 0;NOT 1 1 3.75 500 392 3.75;NOT 1.01 1 3.75 500
659 3.75;
PLF 1 3 2 4 2.5 1;NOT 1 1 7.5 500 988 7.5;NOT 1 1 7.5 500 1865 7.5;
TER 15;
```

```
      SUBROUTINE CONVT
      COMMON IP(10),P(100),G(1000)
      IF(P(1).NE.1.)GOTO100
      F=511./G(4)
      P(6)=F*P(6)
      P(7)=F/P(7)
```

```
 100RETURN
      END

CPLF3LB1                          PLF3 FOR LB1
C     GENERATES HARMONICS WITH ALTERNATING
INSTRUMENTS
C     OPERATES ON NOTE CARDS OF INSTS 1 AND 2
C     P(5) AMPLITUDE,P(6) FREQUENCY ON NOTE CARDS
C     ON PLF CARD, P(4) SPECIFIES HOW MANY FOLLOWING NOTE
CARDS
C     WILL BE OPERATED ON
C     P(5) GIVES THE NUMBER OF HARMONICS GENERATED
C     P(6) SPECIFIES THE TIME SEPARATION BETWEEN
HARMONICS
C     P(7) SPECIFIES THE AMPLITUDE MULTIPLIER FROM ONE
HARMONIC TO THE
C     NEXT
C     P(8) GIVES THEE DURATION DIMINUTION FROM ONE
HARMONIC TO THE NEXT
      SUBROUTINE PLF3
      COMMON IP(10),P(100),D(2000)
      NC=P(4)
      N=P(5)
      IS=P(6)
      FACT=P(7)
      DD=P(8)
      DO 1 I=1,NC
      CALL READ1
      CALL WRITE1(10)
      F=P(6)
      DO 2 J=1,N
```



```
      P(6)=FLOAT(J+1)*F
      P(2)=P(2)+TS
C     TO CHANGE INSTS NUMBER FROM 1 TO 2 AND VICE VERSA
      AINST=P(3)-1.
      IF(AINST)3,3,4
 3    P(3)=2.
      GOTO5
 4      P(3)=1.
 5    CONTINUE
      IF(FACT.GT.0.) P(5)=P(5)*FACT
      P(4)=P(4)-DD
 2    CALL WRITE1(10)
 1    CONTINUE
 100RETURN
      END
```

<u>#502 (Track 35)</u>

This sound results from mixing #500 with itself at different speeds. The speeds have been changed in a way equivalent to playing back #500 simultaneously at a sampling rate of 40,000Hz, 20,000Hz , and 10,000Hz. (This example, in stereo, is again presented backwards.) The remarks mentioned in #512 apply here.

<u>#503 (Track 36)</u>

This sound results from mixing #501 with itself at different speeds. The speeds have been changed in a way equivalent to playing back #501 simultaneously at a sampling rate of 40,000Hz, 20,000Hz, and

10,000Hz. (This example, in stereo, is again presented backwards.) The remarks mentioned in #512 apply here.

#510 (Track 37

This run gives a bunch of siren-like glissandi.

Instrument #1
This instrument delivers a variable frequency sound. The wave is a sine wave with feedback (a process suggested by A. Layzer). The frequency controlling oscillator has a cycle of 8s (P7) repeated 3 times (diagram on following page).







Instrument #2
This instrument gives a noise band with variable center frequency. The 1/2 bandwidth is given by P8. The frequency cycle lasts 6s (P7).





Instrument #3
This instrument gives a wave with variable frequency. The frequency cycle (P7) lasts 12s. The wave given by stored function F2 is truly periodic, but it simulates the sum of inharmonically related partials: the fundamental frequency is 20Hz and the wave consists simply of harmonics #21, 29, 39: thus frequencies 420Hz, 580Hz and 780Hz are present.





F2 is a drastically varying function: to minimize noise due to roundoff errors, IOS is used here (This is the version of the oscillator which interpolates between 2 successive samples whenever the sum of increments is not an integer).

Instrument #4
This instrument gives a sine wave with variable frequency.



Note: here the rate at which the frequency controlling functions are scanned is determined by P7 (converted by P(7)=F/P(7)): it is divorced from the duration of the note; in effect these functions are scanned several times for one note length.


```
COMMENT --------------JCR510------------------------;
COMMENT: SIRENE POUR MUTATION;
COMMENT:TAPE 1779;
COMMENT:FEEDBACK GLISSANDO;
INS 0 1;OSC P6 P7 B4 F5 P8;AD2 B10 P5 B11;
OSC B11 B4 B10 F1 P30;OUT B10 B1;END;
COMMENT:NOISE BAND GLISSANDO;
INS 0 2 ;RAN P5 P8 B4 P30 P29 P28;
OSC P6 P7 B5 F6 P9;OSC B4 B5 B5 F1 P27;OUT B5 B1;END;
COMMENT:INHARMONIC GLISSANDO;
INS 0 3;OSC P6 P7 B4 F7 P8;IOS P5 B4 B5 F2 P30;OUT B5 B1;END;
COMMENT:SINES GLISSANDO;
INS 0 4;OSC P6 P7 B4 F8 P8;OSC P5 B4 B5 F1 P30;OUT B5 B1;END;
SIA 0 4 10000;
GEN 0 2 1 1 1;
GEN 0 4 2 1 21 0 1 512 1 29 0 1 512 1 39 0 1 512;
GEN 0 1 5 .999 0 .999 25 .318 231 .318 281 .999 487 .999 512;
GEN 0 1 6 .377 0 .999 256 .377 512;
GEN 0 1 7 .5 1 .5 15 .9 241 .9 271 .5 497 .5 512;
GEN 0 1 8 .333 0 .333 8 .999 248 .999 264 .333 504 .333 512;
NOT 1 1 24 450 880 8;
NOT 1 2 24 400 1660 6 200;
NOT 1 3 24 200 20 12;
NOT 1 4 24 70 2400 3 0;
NOT 1 4 24 70 2400 3 128;
NOT 1 4 24 70 2400 3 256;
NOT 1 4 24 70 2400 3 384;
TER 25;
```

```
C       SIRENE POUR MUTATION CONVT
      SUBROUTINE CONVT
      COMMON IP(10),P(100),G(1000)
      IF(P(1).NE.1.)GOTO100
      F=511./G(4)
      P(6)=F*P(6)
      P(7)=F/P(7)
      IF(P(3).EQ.2.)P(8)=F*P(8)
  100RETURN
      END
```


#511 (Track 38)

This run gives a bunch of simultaneous glissandi, played at double speed (20,000Hz sampling rate instead of 10,000Hz).

1st Section

Instrument #1

This instrument delivers a variable frequency sound. The wave is a sine wave with feedback (a process suggested by A. Layzer). The frequency controlling oscillator has a cycle of 4.5s, repeated 4 times.



NB: B6 must be reserved for instrument #1

Instrument #2

This instrument gives a noise band with variable center frequency.





Instrument #3

This instrument gives a variable frequency triangular wave.







2nd Section

Instrument #4

This instrument gives a glissando for six "parallel" voices, such that there is a constant frequency difference between the voices (instead of a constant frequency ratio, which would give a constant musical interval). This was first done by J. Clough.



Here the glissando is relatively narrow. The parameters P6, P8, P9, P10, P11, P12 correspond to an initial chord noted:



```
COMMENT: ----------------JCR511--------------------;
COMMENT:GLISSANDI FOR LB;
COMMENT:ON TAPE M2804, FILE 1;
COMMENT:FEEDBACK;
INS 0 1;SET P9;OSC P6 P7 B4 F5 P30;MLT P8 B6 B7;AD2 P5 B7 B8;
OSC B8 B4 B6 F1 P29;OUT B6 B1;END;
COMMENT:NOISE BAND;
INS 0 2;
SET P9;OSC P6 P7 B4 F6 P30;RAN P5 P8 B3 P10 P29 P28;
OSC B3 B4 B5 F1 P27;OUT B5 B1;END;
COMMENT:SIMPLE GLISSANDO;
INS 0 3;
SET P9;OSC P6 P7 B4 F7 P30;OSC P5 B4 B5 F2 P29;OUT B5
B1;END;
SIA 0 4 10000;
GEN 0 2 1 1 1;GEN 0 3 2 0 10 0 -10 0;
GEN 0 1 5 .999 1 .999 50 .85 462 .85 512;
GEN 0 1 6 .999 1 .999 20 .235 492 .235 512;
GEN 0 1 7 .999 1 .999 25 .06 487 .06 512;
NOT 1 1 18 300 208 4.5 .7;
NOT 1 2 16.5 300 440 5.5 80;
NOT 3.75 2 11 300 880 5.5 150;
NOT 1 3 17.6 200 1864 2.2;
NOT 1.7 3 16.9 200 1864 2.2;
NOT 2.4 3 16.2 200 1864 2.2;
SEC 20;
COMMENT:MULTIPLE SYNCHRONOUS GLISSANDI;
INS 0 4;OSC P6 P7 B4 F8 P24;AD2 B4 P8 B5;AD2 B4 P9 B6;AD2 B4
P10 B7;
AD2 B4 P11 B8;AD2 B4 P12 B9;OSC P5 B4 B4 F1 P30;OUT B4 B1;
OSC P5 B5 B5 F1 P29;OUT B5 B1;OSC P5 B6 B6 F1 P28;OUT B6 B1;
```



```
OSC P5 B7 B7 F1 P27;OUT B7 B1;
OSC P5 B8 B8 F1 P26;OUT B8 B1;OSC P5 B9 B9 F1 P25;OUT B9
B1;END;
GEN 0 1 8 .25 1 .25 30 .05 140 .25 200 .25 210 .5 270 .75 290 .5 512;
NOT 1 4 20 300 10.65 20 4.402 9.420 23.09 39.936 84.836;
TER 22;

CGCONVT        CONVT FOR GLISSANDI LB
      SUBROUTINE CONVT
      COMMON IP(10),P(100),G(1000)
      IF(P(1).NE.1.)GOTO100
      F=511./G(4)
      P(7)=F/P(7)
      IF(P(3).EQ.4.)GOTO100
      P(6)=F*P(6)
      IF(P(3).EQ.2.)P(8)=F*P(8)
 100RETURN
      END
```

#512 (Track 39)

This example presents sounds obtained by mixing from the sound of the 2nd section of run #511 (glissandi with constant difference in frequency between voices).

The original sound underwent only transpositions by speed changing before mixing. What differs from one sound of this example to another are both the frequency regions of the sounds (low, medium, high) and the density of mixing, that is, the number of voices. The densest passage has a mixing density of 36, and since the original sound comprises 6 voices, the final sound comprises up to 6x36 i.e., more than 200 voices.

Theoretically sounds of this example could have been obtained directly from the computer, without later manipulation, since the sound manipulations performed electroacoustically (transposition, mixing) are easy to do with MUSIC V. But this process allowed to produce complex textures while saving computer time: and it is quite likely that the sound quality of a computer run comprising such a large number of voices would be very poor, since there are only a few samples for the definition of each voice. Moreover this process allows to control the amplitude balance of the various components of the mixing. It is, of course, subject to well-known inconveniences: noise build-up, synchronization problems (c.f., #490).

#513 (Track 40)

This run presents a little more than one octave of an "endless glissando," which could be pursued indefinitely since it is back to its original point after an octave "descent" (c.f., R.N. Shepard, J.Acoust.Soc.Am.,36, 1964, pp.2346-53; J.C. Risset, J.Acoust.Soc.Am., 46, 1969, p. 88 (abstract only).

The gliding sound comprises 10 components, all generated by instrument #1, diagrammed below



Function F3 controls the frequency of the components. It goes down exponentially from 1 to 2^{-10} (10 octaves below). For each component, the initial sum is specified in P9, the value of these sums for the different components are respectively: 0, \f (1,10) x 511, \f (2,10) x 511, \f (3,10) x 511, . . . \f (9,10) x 511. Since P6, which gives the maximum frequency, has the same value 3900 for all components, the components are initially one octave apart. The duration of the frequency cycle is given in P7 and is 120s: this means that each component goes down an octave after \f (120,12) =12s--and the components stay locked one octave apart. (After one octave descent, the lowest component becomes the highest one).



Function F4 controls the amplitude of the components. It is a bell-shaped curve which consists of a portion of a sine wave with a D.C. bias, if the ordinate scale is in dB. (See description of GEN7). For each component, the initial sum is specified in P8: the value of these sums are the same as those specified in P9, and the duration of the amplitude cycle is the same as that of the frequency cycle. Thus the component amplitudes scan this curve while their frequencies scan the frequency curve. This has the effect of strongly attenuating low and high frequency components. (Even though the specified P8 and P9 are equal, the two oscillators should not share the same P field for the sum). After one octave descent, the pattern is the same as the starting pattern (except for errors due to the imprecise definition of small increments which cause the duration of the cycle to be different from the one expected--this may be severe for less than 36 bit word computers).



Function F1 is a sine wave.

IOS has been used instead of OSC for the three oscillators of the instrument. It gives a truly continuous--not a quantized--frequency glide; similarly it gives a more gradual amplitude change. But it is also preferable for the waveshape oscillator--in this case, as in other cases with glissandi or other frequency modulations, round off errors with OSC (c.f., M. V. Mathews, The Technology of Computer Music, MIT Press, 1969, p.134) are specially noticeable because the corresponding noise goes on and off, diminishing when the frequency is such that the sum of increments (the abscissa) is close to an integer value.

To get continuously descending glissandos, one could compute an entire descent of many octaves; it is more economical to compute one cycle (i.e., one octave) and use the computer to copy these samples successively as many times as desired. However, due to the errors mentioned above, one has to inspect the samples and choose to make the concatenation at a point which will give no appreciable discontinuity in either frequency, amplitude, or waveform.

```
COMMENT:-------------------JCR513-----------------------------;
COMMENT:ENDLESS GLISSANDI WITH 3 IOS;
COMMENT:TAPE M1913;
COMMENT:CYCLE DURATION 12S  10 COMPONENTS;
INS 0 1;IOS P5 P7 B3 F2 P8;IOS P6 P7 B4 F3 P9;
IOS B3 B4 B5 F1 P25;OUT B5 B1;END;
COMMENT:TO SET GENERAL CONVT; SV2 0 10 2 6 -7;
SIA 0 4 10000;
GEN 0 2 1 512 1 1;GEN 0 7 2 0;GEN 0 7 3 -10;
NOT 0 1 14 850 3900 120 0 0;
NOT 0 1 14 850 3900 120 51.1 51.1;
NOT 0 1 14 850 3900 120 102.2 102.2;
NOT 0 1 14 850 3900 120 153.3 153.3;
NOT 0 1 14 850 3900 120 204.4 204.4;
NOT 0 1 14 850 3900 120 255.5 255.5;
NOT 0 1 14 850 3900 120 306.6 306.6;
NOT 0 1 14 850 3900 120 357.7 357.7;
NOT 0 1 14 850 3900 120 408.8 408.8;
NOT 0 1 14 850 3900 120 459.9 459.9;
TER 16;
```


#514 (Track 41)

This run is related to #513: but here, while the components frequencies go down, the center of gravity of the frequency distribution goes up (instead of staying approximately invariant as in #513), so that the sound goes down 3 octaves while becoming shriller--and that it ends up much higher than it started.

The basic instrument, diagrammed below, is similar to that used in #513, except that here an instrument comprises five such units, each of which gives one frequency component; so only two note cards are required to get the 10 components of the sound. Functions F1, F2, and F3 are the same as those used in #513. The initial sums are defined in the same way.



While the component frequencies go down, the spectral envelope goes up because the duration of the entire frequency cycle (given by P8=60s) is longer than the duration of the entire amplitude cycle (given by P7=30s). (This may be easier to understand by examining what happens to the initial spectral configuration of #513 when the amplitude increment is larger than the frequency increment). If the process was allowed to continue longer, the peak of the spectral distribution would continue to be translated towards the highest frequencies and then it would jump to the lowest frequencies and resume its translation upwards.

```
COMMENT: -----------------JCR514--------------------;
COMMENT:TONALITY GOES DOWN  TONE HEIGHT GOES UP;
COMMENT: DURATION 18 S  FREQUENCY CYCLE 6 S;
COMMENT:TAPE M2994 FILE 5;GEN 0 5 4;
INS 0 1;
OSC P5 P7 B2 F2 P10;OSC P6 P8 B3 F3 P11;IOS B2 B3 B2 F1
P30;OUT B2 B1;
OSC P5 P7 B4 F2 P12;OSC P6 P8 B5 F3 P13;IOS B4 B5 B4 F1
P29;OUT B4 B1;
OSC P5 P7 B6 F2 P14;OSC P6 P8 B7 F3 P15;IOS B6 B7 B6 F1
P28;OUT B6 B1;
OSC P5 P7 B8 F2 P16;OSC P6 P8 B9 F3 P17;IOS B8 B9 B8 F1
P27;OUT B8 B1;
OSC P5 P7 B10 F2 P18;OSC P6 P8 B11 F3 P19;IOS B10 B11 B10 F1
P26;
OUT B10 B1;END;
SIA 0 4 10000;
GEN 0 2 1 1 1;GEN 0 7 2 0;GEN 0 7 3 -10;
COMMENT;FREQUENCY CYCLE 6 S  AMPLITUDE CYCLE 3 S;
NOT 0 1 18 500 4000 30 60 0 0 0 51.1 5151 102.2 102.2 153.3 153.3
```



```
204.4 204.4;
NOT 0 1 18 500 4000 30 60 0 255.5 255.5 306.6 306.6 357.5 357.5
408.8 408.8 459.9 459.9;
TER 20;

C      CONVT POUR CONFLIT CHROMA HAUTEUR BRUTE
       SUBROUTINE CONVT
       COMMON IP(10),P(100),G(1000)
       IF(P(1).NE.1.)GOTO100
       F=511./G(4)
       P(6)=F*P(6)
       P(7)=F/P(7)
       P(8)=F/P(8)
 100
       RETURN
       END
```

#515 (Track 42)

This run presents sounds whose tone height goes up (or down) continuously, without an octave jump, while their tonality remains invariant (in this case corresponding to a B). This is achieved by having fixed frequency octave components whose spectral envelope is translated as in #514.I

Instrument #1 is used for each of the 8 components of the sounds. It is diagrammed below. The component frequency is given by P6; all components are in octave relation. For each component, the initial sum is specified in P8; the value of these sums for the different components are close to, respectively, 0, \f (1,8) x511, \f (2,8) x 511, . . . , \f (9,10) x 511.

Each sound lasts 5s, which corresponds to less than an entire amplitude cycle.



In the 1st section, function F2 is a single peak bell-shaped curve with 84dB difference between peak and end points ordinates (c.f., description of GEN7).

In the 2nd section, function F2 is a single peak bell-shaped curve with 42dB difference between peak and end points ordinates (for this section and the two which follow, c.f., description of GEN8).

In the 3rd section, function F2 1s a double peak bell-shaped curve--hence the repetition of the pattern.

In the 4th section, function F2 is a triple peak bell-shaped curve.

Note: effects similar to those obtained here can probably be obtained more economically, if not as conveniently and precisely, through the use of FLT. This remark also applies to #516.

```
COMMENT: --------------------JCR515-------------------------------;
COMMENT:SPECTRAL ENVELOPE TRANSLATION FOR OCTAVE
COMPONENTS;
COMMENT:FIXED FREQUENCIES;
COMMENT:TAPE 1669 FILE 2;GEN 0 5 1;
INS 0 1;OSC P5 P7 B3 F2 P8;OSC B3 P6 B3 F1 P25;OUT B3 B1;END;
SIA 0 4 10000;
COMMENT:TO SET GENERAL CONVT;  SV2 0 10 1 6;
GEN 0 2 1 1 1;
COMMENT:AMPLITUDE FUNCTION ONE PEAK 84 DB AMBITUS;
GEN 0 8 2  0;
NOT 1 1 5 500 30 .00716  128;
NOT 1 1 5 500 60 .00716 192;
NOT 1 1 5 500 120 .00716 256;
NOT 1 1 5 500 240 .00716 320;
NOT 1 1 5 500 480 .00716 384;
NOT 1 1 5 500 960 .00716 448;
NOT 1 1 5 500 1920 .00716 0;
NOT 1 1 5 500 3840 .00716 64;
SEC 7;
```

```
COMMENT: ONE PEAK AMBITUS 42 DB;  GEN 0 8 2 0;
NOT 1 1 5 500 30 .00716  128;
NOT 1 1 5 500 60 .00716 192;
NOT 1 1 5 500 120 .00716 256;
NOT 1 1 5 500 240 .00716 320;
NOT 1 1 5 500 480 .00716 384;
NOT 1 1 5 500 960 .00716 448;
NOT 1 1 5 500 1920 .00716 0;
NOT 1 1 5 500 3840 .00716 64;
SEC 7;
COMMENT: TWO PEAKS AMBITUS 42 DB;  GEN 0 8 2 1;
NOT 1 1 5 500 30 .00716  128;
NOT 1 1 5 500 60 .00716 192;
NOT 1 1 5 500 120 .00716 256;
NOT 1 1 5 500 240 .00716 320;
NOT 1 1 5 500 480 .00716 384;
NOT 1 1 5 500 960 .00716 448;
NOT 1 1 5 500 1920 .00716 0;
NOT 1 1 5 500 3840 .00716 64;
SEC 7;
COMMENT: THREE PEAKS AMBITUS 42 DB;  GEN 0 8 2 -1;
NOT 1 1 5 500 30 .00716  128;
NOT 1 1 5 500 60 .00716 192;
NOT 1 1 5 500 120 .00716 256;
NOT 1 1 5 500 240 .00716 320;
NOT 1 1 5 500 480 .00716 384;
NOT 1 1 5 500 960 .00716 448;
NOT 1 1 5 500 1920 .00716 0;
NOT 1 1 5 500 3840 .00716 64;
TER 7;
```


#516 (Track 43)

This run presents sounds of variable spectrum; the variation of spectrum is achieved by translating (as in #514) the spectral envelope of fixed frequency components.

The instrument used is the same as in #515. However, 10 frequency components are used instead of 8, and here they are not in octave relation.

In the first 3 sections the frequency components form a harmonic series:

(1) In section 1, function F2 is a single peak bell-shaped curve with 42 dB difference between peak and end points ordinates (For the amplitude controlling functions of this run, c.f. description of GEN8). (2) In section 2, F2 is a double peak bell-shaped curve.

(3) In section 3, F2 is a triple peak bell-shaped curve.

In the last 3 sections the frequency components are not harmonically related.:

(4) In section 4, F2 is as in section 1.

(5) In section 5, F2 is as in section 2.

(6) In section 6, F2 is as in section 3.

See note at the end of #515.

```
COMMENT: --------------------JCR516--------------------------------;
COMMENT:VARIABLE SPECTRUM SOUNDS THROUGH SPECTRAL
ENVELOPE TRANSLATION;
COMMENT:FIRST HARMONIC  THEN INHARMONIC FREQUENCIES;
COMMENT:TAPE M1485;
INS 0 1;OSC P5 P7 B3 F2 P8;OSC B3 P6 B3 F1 P25;OUT B3 B1;END;
SIA 0 4 10000;
```

```
COMMENT:TO SET GENERAL CONVT;  SV2 0 10 1 6;
GEN 0 2 1 1 1;
COMMENT:HARMONIC FREQUENCIES;
COMMENT:AMPLITUDE FUNCTION ONE PEAK 84 DB AMBITUS;
GEN 0 8 2  0;
NOT 1 1 3 500 200  .02 0;
NOT 1 1 3 500 400  .02 51.1;
NOT 1 1 3 500 600  .02 102.2;
NOT 1 1 3 500 800   .02 153.3;
NOT 1 1 3 500 1000 .02 204.4;
NOT 1 1 3 500 1200 .02 255.5;
NOT 1 1 3 500 1400 .02 306.6;
NOT 1 1 3 500 1600 .02 357.7;
NOT 1 1 3 500 1800 .02 408.8;
NOT 1 1 3 500 2000 .02 459.9;
SEC 7;
COMMENT: TWO PEAKS;  GEN 0 8 2 1;
NOT 1 1 3 500 200  .02 0;
NOT 1 1 3 500 400  .02 51.1;
NOT 1 1 3 500 600  .02 102.2;
NOT 1 1 3 500 800   .02 153.3;
NOT 1 1 3 500 1000 .02 204.4;
NOT 1 1 3 500 1200 .02 255.5;
NOT 1 1 3 500 1400 .02 306.6;
NOT 1 1 3 500 1600 .02 357.7;
NOT 1 1 3 500 1800 .02 408.8;
NOT 1 1 3 500 2000 .02 459.9;
```



```
SEC 7;
COMMENT: THREE PEAKS;  GEN 0 8 2 -1;
NOT 1 1 3 500 200  .02 0;
NOT 1 1 3 500 400  .02 51.1;
NOT 1 1 3 500 600  .02 102.2;
NOT 1 1 3 500 800   .02 153.3;
NOT 1 1 3 500 1000 .02 204.4;
NOT 1 1 3 500 1200 .02 255.5;
NOT 1 1 3 500 1400 .02 306.6;
NOT 1 1 3 500 1600 .02 357.7;
NOT 1 1 3 500 1800 .02 408.8;
NOT 1 1 3 500 2000 .02 459.9;
SEC 7;
COMMENT:INHARMONIC FREQUENCIES;
COMMENT:AMPLITUDE FUNCTION ONE PEAK 84 DB AMBITUS;
GEN 0 8 2  0;
NOT 1 1 3 500 80 .02 255.5;
NOT 1 1 3 500 230 .02 459.9;
NOT 1 1 3 500 315 .02 306.6;
NOT 1 1 3 500 650 .02 204.4;
NOT 1 1 3 500 750  .02 102.2;
NOT 1 1 3 500 930  .02 0;
NOT 1 1 3 500 1400   .02 153.3;
NOT 1 1 3 500 2500 .02 357.7;
NOT 1 1 3 500 300 .02 408.8;
SEC 7;
COMMENT: TWO PEAKS;  GEN 0 8 2 1;
NOT 1 1 3 500 80 .02 255.5;
NOT 1 1 3 500 230 .02 459.9;
NOT 1 1 3 500 315 .02 306.6;
NOT 1 1 3 500 650 .02 204.4;
```

```
NOT 1 1 3 500 750  .02 102.2;
NOT 1 1 3 500 930  .02 0;
NOT 1 1 3 500 1400   .02 153.3;
NOT 1 1 3 500 2500 .02 357.7;
NOT 1 1 3 500 300 .02 408.8;
SEC 7;
COMMENT: THREE PEAKS;  GEN 0 8 2 -1;
NOT 1 1 3 500 80 .02 255.5;
NOT 1 1 3 500 230 .02 459.9;
NOT 1 1 3 500 315 .02 306.6;
NOT 1 1 3 500 650 .02 204.4;
NOT 1 1 3 500 750  .02 102.2;
NOT 1 1 3 500 930  .02 0;
NOT 1 1 3 500 1400   .02 153.3;
NOT 1 1 3 500 2500 .02 357.7;
NOT 1 1 3 500 300 .02 408.8;
TER 7;
```

#517 (Track 44)

This example presents a fragment obtained by mixing from runs #510, 511, 513 to 516 (and a couple of other similar runs).

The original sounds underwent only transpositions by speed changing before mixing (except for the sound analogous to the one presented in #514, which was artificially reverberated by means of an EMT metallic plate--a similar reverberation could have been performed by computer).

The remarks mentioned by #512 also apply here.

#550 (Track 45)

This run presents an attempt to prolong harmony into timbre: a chord, played with a timbre generated in a way similar to ring modulation, is echoed by a gong-type sound whose components are the fundamentals of the chord. The latter sound is perceived as a whole rather than as a chord, yet its tone quality is clearly related to the chord's harmony. The passage is as follows:



Instrument #1 (below) is used to generate the notes of the chord, in a way similar to a ring modulator combining the outputs of a sine wave and a square wave oscillator. Low values have to be used for the amplitude inputs P5 and P7, since the resulting maximum amplitude will be of the order of P5 x P7. The dominant frequency of a note played with this instrument is the difference between P8 and P6. Function F3 controls the envelope of the sine wave component, hence the envelope of the note; this modulation at the same time produces spectrum changes. Notes of the chord are first played with a short (10 ms), percussive attack, then with a cresc-decresc type envelope.



Instrument #2, used for the gong-like sound, is similar to instrument #1 of #420: there is one note card for each frequency component; the waveshape is a sine wave; each component is decaying exponentially at its own rate. As was mentioned earlier, the frequencies of the components are equal to the frequencies of the notes of the preceding chord.

```
COMMENT: -----------------------JCR550-------------------------;
COMMENT:PROLONGATION OF HARMONY INTO TIMBRE;
INS 0 1;ENV P5 F3 B3 P9 P10 P11 P30;OSC B3 P6 B3 F1 P29;
OSC P7 P8 B4 F2 P28;MLT B3 B4 B3;OUT B3 B1;END;
INS 0 2;OSC P5 P7 B3 F4 P30;OSC B3 P6 B3 F1 P29;OUT B3
B1;END;
COMMENT:TO SET GENERAL CONVT;
SV2 0 10 3 6 8 109;
SV2 0 20 2 6 -7;
GEN 0 2 1 1 1;
GEN 0 3 2 0 10 10 10 10 10 0 -10 -10 -10 -10 -10 0;
GEN 0 6 3 10 .99 .99 10;
GEN 0 7 4 -9;
NOT .5 1 .6 18 424 18 1000 0.1 0 .6;
NOT .6 1 .6 18 727 18 1000 .01 0 .6;
NOT .9 1 3.6 18 424 18 1000 2.3 0 1.2;
NOT .9 1 .6 18 1542 18 2000 .01 0 .6;
NOT 1 1 3.5 18 727 18 1000 3.2 0 1.2;
NOT 1.1 1 .6 18 1136 18 2000 .01 0 .6;
NOT 1.3 1 3.2 18 1542 18 2000 1.9 0 1.2;
NOT 1.4 1 .6 18 1342 18 2000 .01 0 .6;
NOT 1.5 1 3 18 1136 18 2000 1.9 0 1.2;
NOT 1.8 1 2.7 18 1342 18 2000 1.4 0 1.2;
COMMENT:TIMBRE ECHO TO PREVIOUS HARMONY;
```

```
NOT 4 2 10 400 273 10;NOT 4 2 7.5 200 455 7.5;
NOT 4 2 4.5 200 576 4.5;NOT 4 2 6.5 150 648 6.5;
NOT 4 2 4 150 864 4;
TER 15;
```